
要約: 

 本稿は、伊藤(2003, 第 1 章)の「部品調達問題」についての数値計算モデルを構築し、契約理論モ

デルの特性について数値例によって理解を深めようとするものである。前半では伊藤に従って 2 タイプ

と 3 タイプのケースを考察し、後半では理論モデルに導入された簡単化のための仮定を外した場合や、

簡単化の仮定が成立する状況がどの程度起こりそうであるかについても検討する。さらに、数値計算モ

デルの優位性を生かして、タイプ数が非常に多い場合であってもモデルを容易に拡張・適用できること

を示す。 

1. プログラミング･モデル・アプローチについて 

 本稿は、伊藤(2003)の第 1 章「アドバース･セレクション」が示すプリンシパル=エージェント問題(部品

調達問題)を数値的に解く手法について論じる。同書は、契約理論を一般的かつ厳密に展開している

ために、初学者にとって理解しにくいところがある。そのため、そこで示されている問題を数値問題として

作って解いていくことによって、契約理論をよりよく理解しようとするのが、本稿の目的である。 

 具体的には、メーカー(プリンシパル)が、複数タイプ存在するサプライヤー(エージェント)と部品供給契

約を結ぶ問題を考える。伊藤の議論に従って、順次、サプライヤーのタイプが 2 個のモデルと 3 個のモ

デルに対応したコンピュータ・プログラムを作っていく。それぞれの場合について、情報の非対称性のな

いファーストベスト・モデルと、それがあるセカンドベスト・モデルのプログラムを作る。セカンドベスト･モ

デルは、サプライヤーが自分以外のサプライヤーであると偽る可能性があり、これを防ぐための誘因両

立制約が必要となる。タイプ数が増えるにつれて、誘因両立制約が複雑になる。この複雑性を避けるた

めに、しばしば理論分析ではいくつかの仮定を設けて理論モデルの簡単化を図る。本稿では、こうした

簡単化の仮定を課した場合はもとより、これらが成立しない場合についても数値例を用いて考察する。

さらに、数値計算モデルの優位性を生かし、タイプ数が非常に多くなった(たとえば、10 個)場合につい

ても検討する。 

数値モデルを解くために、数値計算ソフトウェアとして、GAMS(General Algebraic Modeling 

System)を用いることにし、そのためのプログラムを提示する。1GAMS の文法に関しては、本稿中のプ

ログラムに即して最低限の説明を施しているが、より詳細な文法については細江他(2004, 第 3 章)が

簡便で有用であろう。各自が GAMS 試用版を使って、本稿で示すプログラムをそのまま、あるいは、自

由にパラメータなどを変更して、契約理論の要諦について数値例を用いて「実感」することができる。 

                                                      

1 GAMS は市販のソフトウェアであるが、その試用版は、解くことができるモデルの大きさに制限があるものの、

GAMS Development Corporation の Web サイト(http://www.gams.com/dowmload/)からダウンロードして無

料で利用できる。なお、本稿で取り扱う問題は、この試用版で解くことができる程度に小さい問題であるので、この

試用版で十分である。 

 もちろん、種々の特定化を施された関数型や、仮定されたパラメータ設定の下で作られた数値例であ

る以上、その結果から一般的な結論を導くことには慎重にならざるを得ない。技術的には、数値計算モ

デルを構築するために新しいコンピュータ言語を習得する手間も小さくはない。その一方で、理論モデ

ルのエッセンスを数値例を用いて表現するわれわれの手法—プログラミング･モデル・アプローチと呼ぼ

う—によって、わかりやすさという恩恵を得ることができる。また、一般的な設定の下で理論的検討を行う

ために導入されたいくつかの技術的な仮定から解放されるという利点もある。われわれとしては、読者に

よって本稿の教育的な面での有用性が見いだされることを期待するものである。 

 

2. アドバース・セレクション：部品調達問題 

 本稿では、伊藤の第 1 章で示されている部品調達問題について考察する。この問題はアドバース・セ

レクションを考える際の最も基本的な問題である。すなわち、1 台の最終生産物を作るメーカー(プリンシ

パル)がその生産に必要な部品をサプライヤー(エージェント)から購入することを考える。サプライヤーは

自分自身の技術水準(効率性に関して、2 または 3 タイプあるとしよう)を私的情報として持っているが、

メーカーには観察できない。こうした状況の下で、メーカーは自己の効用を最大にする契約を策定する。 

本稿で扱うものは、タイプが離散変数のケースであり、2.1 では、サプライヤーのタイプを 2 個(効率的

タイプと非効率的タイプ)とする。そこで、伊藤 1.1.2 に従って、ベンチマークのために、まず、非対称情

報がないものとしてファーストベスト(最善)の均衡を解き、つぎに、伊藤 1.1.4 に従って、非対称情報が

あるとしてセカンドベスト(次善)の均衡を解くことにする。ファーストベスト・モデルとしては、最初に、契約

を結ぼうとしているサプライヤーが効率的であることを、メーカーが知った上で契約する問題のモデルと、

非効率的であると分かった上で契約する問題のモデルを個別に作成して解く。つぎに、これらの 2 つの

モデルを 1 つのモデルとして解く。続いて、セカンドベスト・モデルを提示する。ここでは、サプライヤー

のそれぞれが、自分以外の全てのサプライヤーのどれかであると偽る可能性があり、これを防ぐための

(大域的)誘因両立制約を組み込まなければいけない。この制約のために、サプライヤーはファーストベ

スト以下の効用しか実現できない。 

つぎはタイプが 3 個の場合である。これは、2.2 および 2.3 で検討する。サプライヤーのタイプが 2

個から一般に N 個に増えたとき、複雑になるのが誘因両立制約である。申告する可能性のあるタイプ

数が多くなると、誘因両立制約の数が急速に増えることになる。そこで、追加的仮定—Spence-

Mirrlees の単一交差性(single crossing property, SCP)や単調性—を加えて簡単化を図る。(SCP

や単調性については、2.2.3 で詳しく説明する。)これらの仮定はモデルを簡単化して、最適契約の性

質を解析的に明らかにすることを可能にする。一方、本稿では、これらの簡単化のための仮定をしない

場合も考える。さらに、タイプが 3 個より多いモデルも提示する。プログラミング技術上、この種の拡張が

きわめて容易であることが示される。また、この種の考察を通じて、タイプ数が増えるとこれらの簡単化の

ための仮定が成り立つ場合が限られてくることも示唆される。 

 なお、伊藤のオリジナル・モデルと本稿の数値計算モデルの間には(また、伊藤の 1.1 と 1.3 の間に

も)、関数や変数に関する仮定や記号表記に違いがあるので、その相違点を表 2.1 にまとめておく。 

 

表 2.1: 関数・変数に関する仮定の相違 

関数・変数 伊藤(2003)における仮定 本稿における仮定 

( )⋅b  ( ) 00 =b , 

任意の xx < に対して 

( ) 0' >xb , ( ) +∞=0'b , ( ) 0' =xb  

任意の xに対して 

( ) 0'' <xb  

(ただし、 [ ]xXx ,0=∈ ) 

( ) 5.0xxb =  

(厳密には ( ) 0' =xb を満たさないが、近似

として十分) 

( )⋅ic  

または ( )⋅u  

-2 タイプ・モデル: 

( ) iiii xxc θ=  

- N タイプ・モデル( 3≥N ): 

( )iixu θ, は iθ の厳密な増加関数 

( ) iiii xxc θ=  

( ) iiii xxu θθ −=,  

( ( )iixu θ, は iθ の厳密な減少関数) 

タ イ プ i と

iθ  

-2 タイプ・モデル: 

100 θθ <<  

0θ が最も効率的 

- N タイプ・モデル( 3≥N ): 

Nθθθ <<< ...10  

Nθ が最も効率的 

-2 タイプ・モデル: 

- N タイプ・モデル( 3≥N ): 
















=

















3.0

2.0

1.0

2

1

0

θ
θ
θ

 

0θ が最も効率的 

ru ( 留 保

効用) 

陽表的に考慮しない(=0) ru を導入。ただし、本文中では 0=ru 。 

メーカーの

効用 
-伊藤 1.1 

( ) ( )[ ]

( )[ ]∑

∑
−=

−

i

iiii

i

iiii

xxbp

xcxbp

θ
 

-伊藤 1.3 

( ) ( )
( )[ ] [ ]

( )[ ]∑

∑
−=

−−−=

−

i

iiii

i

iiiiii

iii

xxbp

xwwxbp

UxS

θ

θ

θθ,

 

 

( ) ( )[ ]

( )[ ]∑

∑
−=

−

i

iiii

i

iiii

xxbp

xcxbp

θ
 

 

2.1 2 タイプ・モデル 

サプライヤーのタイプを添え字 iで表し、効率的(伊藤では 0=i 、ここでは eff)と非効率的(同じく

1=i 、ここでは inf)の 2 種類があるとする。効率性は、それらの限界費用( iθ , theta(i))の高低に

よって決められるとする。メーカーが作る契約は、サプライヤーが自分自身のタイプを「 i (効率的、また

は、非効率的)である」と申告するのに対して、メーカーの効用を最大化するようなメカニズム—部品の品

質 ix とサプライヤーに支払うべき価格 iw のペア—を決めるものである。メーカーがサプライヤーのタイ

プを知っている(情報の非対称性なしの)場合には、サプライヤーが情報の非対称性を利用して自己に

有利な申告をすることがないので、ファーストベストが実現できる。一方、メーカーがサプライヤーのタイ

プを知らない(情報の非対称性有りの)場合には、サプライヤーに偽りの申告をする気を起こさせない、

すなわち、偽りの報告をしてもトクにならないような契約を提示する必要がある。そのような契約が、情報

の非対称性に伴う非効率を少なくするという意味で最適な契約である。もちろんこの場合には、メーカー

は、ファーストベストと同じ効用水準を達成できるわけではなく、セカンドベストの効用水準しか実現でき

ない。2 

 

                                                      

2 伊藤の第 1 章では、利益と効用が互換的に用いられているが、ここでは効用だけを用いる。 

2.1.1 ファーストベスト・モデル: 2 種類のサプライヤー別のモデル 

 部品調達問題のファーストベスト・モデル(あるいは、ベンチマーク・モデル)は、メーカーが、サプライヤ

ーの提供する部品製造の限界費用( iθ , theta(i))を知った上で、部品の品質( ix )とサプライヤーに

支払うべき価格( iw )を決めた契約を作るものである。ここでは、伊藤 1.1.2 に従う形で、ファーストベス

ト･モデルを作る。すなわち、サプライヤーが効率的(eff)な場合の問題と、非効率的(inf)な場合の問

題を、別々の問題として考える。各モデルは、単純に決定変数 ix が非負であるという条件の下での、メ

ーカーの効用—これは非対称情報の問題がないために、総余剰に等しくなる—最大化問題である。 

 

2.1.2 効率的サプライヤーのモデル (PS2_F_eff.gms) 

はじめに、効率的サプライヤーのモデルの入力ファイル(PS2_F_eff.gms)をみていこう。このモデ

ルは、伊藤 1.1.2 のモデルを基に、表 2.1 にまとめように一部の関数を特定化して書き換えたものであ

る。 

 ( ) ( )[ ]∑ −=
i

iii xcxbUtilmax      (2.1) 

subject to 

 ( ) 5.0

ii xxb =  i∀       (2.2) 

 ( ) 0=− iiii xxc θ  i∀       (2.3) 

一般には、 infeffi ,= であるが、ここでは、 effi = (効率的サプライヤー)のみについて問題を考えて

解く。リスト 2.1 に、そのための入力ファイルを示す。そこでは、左端に行番号が付されているが、これは

説明の便宜のためであって、実際には入力しない(入力するとエラーになる)。 

 

リスト 2.1: 効率的サプライヤー・モデルの入力ファイル (PS2_F_eff.gms) 

...  

7 * Definition of Set 

8 Set     i       type of supplier        /eff/; 

9 * Definition of Parameters 

10 Parameter 

11         theta(i)        efficiency      /eff    0.2/; 

12 * Definition of Primal/Dual Variables 

13 Positive Variable 

14         x(i)            quality 

15         b(i)            maker's revenue 

16         c(i)            cost; 

17 Variable 

18         Util            maker's utility; 

19 Equation 

20         obj             maker's utility function 

21         rev(i)          maker's revenue function 

22         pc(i)           participation constraint; 

23 * Specification of Equations 

24 obj..   Util =e= sum(i, (b(i)-c(i))); 

25 rev(i)..b(i) =e= x(i)**(0.5); 

26 pc(i).. c(i)-theta(i)*x(i) =e= 0; 

27  

28 * Setting Lower Bounds on Variables to Avoid Division by Zero 

29 x.lo(i)=0.0001;; 

30  

31 * Defining and Solving the Model 

32 Model FB1 /all/; 

33 Solve FB1 maximizing Util using NLP; 

34  

35 Parameter 

36                 db(i)   derivative of b 

37                 w(i)    price 

38 ; 

39 db(i)   =0.5*x.l(i)**(-0.5); 

40 w(i)    =c.l(i); 

41  

42 Display x.l, b.l, c.l, util.l, db, w; 

43 * End of Model 

 

 プログラムの 8 行目にサプライヤーのタイプを表す添え字 i が、指示子(要は命令のようなもの)Set

で定義されている。効率的サプライヤー(だけ)のモデルでは、効率的サプライヤーを表す eff だけが

あれば十分なため添え字 i を省略しても構わないが、この後で、この同じコンピュータ・モデルを基に、

非効率的サプライヤーのモデルを作ることを考えてこの命令がある。11 行目は、eff サプライヤーの限

界費用 theta(i)を定義して、その値を 0.2 と設定(伊藤第 1 章にはない本稿における独自の仮定)

している。 

13–18 行目は、GAMS プログラム内で用いる決定変数(内生変数)の名称を定義している。理論モデ

ル(伊藤のオリジナル)における決定変数は、部品の品質 ix だけであるが、メーカーの収入関数 ( )⋅b と

費用関数 ( )⋅ic を数値計算のために特定化するために、x(i)以外にそれらの値 b(i), c(i)がコンピ

ュータ・プログラム内の決定変数としてここで定義されている。さらに、x(i), b(i), c(i)は非負である

として、Positive Variable という指示子で定義する。3もう 1 つの変数 Util は、これも伊藤第 1

章では陽表的に用いられていないが、このプログラムにおける目的関数の値の名称として定義する。目

的関数の値の符号(値域)はモデルを解く前には分からないので、正負両方を取り得る Variable とす

る。19–22 行目は、目的関数(メーカーの効用関数)と制約式(収入関数、費用関数)の名称である。

GAMS プログラム中では、制約式(目的関数を含む)名称を与えなければならない。4obj は目的関数

(2.1)、rev(i)は収入関数(2.2)、pc(i)は費用関数(2.3)の名称であるとした。 

24–26 行目が、最大化問題を構成する制約式(目的関数を含む)である。24 行目が目的関数である。

525 行目で、メーカーの収入 b(i)が、x(i)の関数として定義されている。収入関数 ( )⋅b について、伊

藤では一般的な形で仮定されているが、ここでは、その最も簡単な関数の代表例として ( ) 5.0

ii xxb = を

仮定する。なお、GAMS では、「=e=」は通常の等号、(行頭以外の)「*」は乗算、「**」はベキ乗を意味

する。26 行目で、伊藤 1.1.2 の通りに、サプライヤーの費用 ( )⋅ic を限界生産費 iθ が一定の一次関数と

する。 

 29 行目は、数値計算の途中で x(i)がゼロになって計算に支障が起こるのを防ぐために、x(i)の下

限として任意の、しかし、計算結果には影響を与えない非常に小さい正数を与えるものである。もちろん、

x(i)の解がここで与えた下限に一致してしまった場合は、この人為的な下限の設定を再考する必要が

                                                      

3 なお、GAMS プログラム中では、Positive Variable は非負の変数の意味であり、ゼロも含むことに注意。 

4 通常であれば、「(1)式」, 「(2-3)式」といった式番号を用いるが、それと同様で、しかし、より自由度の高い式の名

前を与えることができる。 

5 すべての状況 i に対応した収入と費用の差額の総額がメーカーの効用であるので、i についての和 iΣ (GAMS

プログラム中では、「sum(i, ...)」)をとってある。 

ある。632 行目は、これまでに定義されたすべて(all)の制約式(目的関数を含む)で構成されるモデル

の名前を FB1 と定義し、33 行目は、このモデル FB1 を、非線形計画問題(nonlinear programming, 

NLP)ソルバーを使って、Util を最大化するように解くという指示である。7 

 35 行目以降では、モデルの数値解を基に、さらなる分析をするためのものである。db(i)は、関数

b(i)を x(i)の解 x.l(i)において評価した一階微分の値である。契約は、部品の品質 x(i)ととも

にメーカーがサプライヤーに支払う価格 w(i)を提示するものであるので、ここで w(i)を計算する(た

だし、プログラムからわかるように、これは実質的に単なる代入である)。ファーストベストに

おいては、契約締結時の交渉力はメーカー側にあるとしているので、w(i)はつねに生産費用 c(i)

に等しい。そこで w(i)の値を c(i)の解の値に等しくしている。 (モデルの解を計算に使う場合は、変

数の後に「.l」(数字の 1 でなくアルファベットのエル)を付す。たとえば、x.l(i)のように。また、制約

つき最大化問題のラグランジュ乗数の解を使う場合は、「.m」を付す。たとえば、pc.m(i)というように。)

なお、最大化問題を構成する制約式(目的関数を含む)以外では、等号には「=e=」ではなく、普通の

「=」を使う。842 行目は、モデルの解(ラグランジュ乗数を含む)や上で計算した数値を表示 Display

せよという指示である。 

 最後に、GAMS のプログラムにおいて、「*」(アスタリスク)で冒頭が始まる行は単なるメモである。プロ

                                                      

6 GAMS プログラム中で先験的に設定する変数の下限については、細江他(2004, 第 3 章)。 

7 GAMS は様々な種類の問題((非)線形計画問題、整数計画問題、相補計画問題等)を解くことができるように、そ

れぞれの問題に応じたソルバーを備えている。プログラムのこの部分では、どのソルバーを用いて問題を解くべきか

を指示している。 

8 ラグランジュ乗数は、最大化問題が、 

 ( )xfmax  

subject to  ( ) 0≤ii xg  i∀  

という形式で表されるとき、そのラグランジュ乗数が非負で表現される(Varian (1992, 第 27 章))。しかし、本稿中

のプログラムは、プログラム自体の解釈のしやすさを重視して、必ずしもそのような形で表されていない。したがって、

ラグランジュ乗数の計算結果が負で表現される場合があるが、本質的に何ら相違はない。 

グラム自体に何ら影響を与えない一方で、このメモがあることで他の人がプログラムの内容を、よりよく理

解できたり、モデル作成者自身の記憶の助けになったりもする。(時間が経つと、モデル作成者自身でさ

え、自分の書いたものを思い出せないことがある。)可能な限りメモを書いておくことが GAMS プログラ

ム作成上のポイントである。 

 以上が入力ファイルの中身である。こうしたコンピュータ・プログラムを GAMS IDE と呼ばれるソフトウ

ェア上で作成して、メニューバー上のアイコン クリックすると、GAMS がこのコンピュータ・プログラム

を入力ファイルとして解釈して計算を行い、出力ファイル(PS2_eff.lst, リスト 2.2)が表示される。9リ

スト 2.2 では、左端に行番号が示されているが、入力ファイル(リスト 2.1)と同様に、説明の便宜のためで

ある。 

 

リスト 2.2: 効率的サプライヤー・モデルの出力ファイル(抜粋)  (PS2_eff.lst) 

...    

11    7  * Definition of Set 

12    8  Set     i       type of supplier        /eff/; 

13    9  * Definition of Parameters 

14   10  Parameter 

15   11          theta(i)        efficiency      /eff    0.2/; 

16   12  * Definition of Primal/Dual Variables 

17   13  Positive Variable 

18   14          x(i)            quality 

...    

74                S O L V E      S U M M A R Y 

75  

76      MODEL   FB1                 OBJECTIVE  Util 

77      TYPE    NLP                 DIRECTION  MAXIMIZE 

78      SOLVER  CONOPT              FROM LINE  33 

79  

80 **** SOLVER STATUS     1 NORMAL COMPLETION          

81 **** MODEL STATUS      2 LOCALLY OPTIMAL            

82 **** OBJECTIVE VALUE                1.2500 

...    

102  ** Optimal solution. Reduced gradient less than tolerance. 

                                                      

9 GAMS および GAMS IDE の利用方法については、細江他(2004, 第 3 章および付録 C)、または、Brooke et 

al. (2010)、および、より技術的な点は McCarl (2009)参照。 

...    

115                        LOWER     LEVEL     UPPER    MARGINAL 

116  

117 ---- EQU obj             .         .         .        1.000       

118  

119   obj  maker's utility function 

120  

121 ---- EQU rev  maker's revenue function 

122  

123        LOWER     LEVEL     UPPER    MARGINAL 

124  

125 eff      .         .         .        1.000       

126  

127 ---- EQU pc  participation constraint 

128  

129        LOWER     LEVEL     UPPER    MARGINAL 

130  

131 eff      .         .         .       -1.000       

132  

133 ---- VAR x  quality 

134  

135        LOWER     LEVEL     UPPER    MARGINAL 

136  

137 eff 1.0000E-7     6.250     +INF       EPS        

138  

139 ---- VAR b  maker's revenue 

140  

141        LOWER     LEVEL     UPPER    MARGINAL 

142  

143 eff      .        2.500     +INF       .          

144  

145 ---- VAR c  cost 

146  

147        LOWER     LEVEL     UPPER    MARGINAL 

148  

149 eff      .        1.250     +INF       .          

150  

151                        LOWER     LEVEL     UPPER    MARGINAL 

152  

153 ---- VAR Util           -INF      1.250     +INF       .          

...    

167 ----     42 VARIABLE x.L  quality 

168  

169 eff 6.250 

170  

171  

172 ----     42 VARIABLE b.L  maker's revenue 

173  

174 eff 2.500 

175  

176  

177 ----     42 VARIABLE c.L  cost 

178  

179 eff 1.250 

180  

181  

182 ----     42 VARIABLE Util.L           =    1.250  maker's utility 

183  

184 ----     42 PARAMETER db  derivative of b 

185  

186 eff 0.200 

187  

188  

189 ----     42 PARAMETER w  price 

190  

191 eff 1.250 

...    

 

 この出力ファイルを解釈する。出力ファイルの前半部には、入力ファイルの中身が行番号つきでその

まま表われる。その先にある SOLVE SUMMARY に、「** Optimal solution」の表示があることを

確認する。(この表示がない場合には、以下にどのような数値が表されていても、解としては意味を持た

ない。)GAMS の解の出力には EQU(おもに制約式のラグランジュ乗数の値を表示する)と VAR(おもに

決定変数(内生変数)の値を表示する)の 2 種類がある。はじめに EQU ブロックが表示される。EQU ブロ

ックの MARGINAL の列に、それぞれの式のラグランジュ乗数が示されている。たとえば、制約式

rev(i)のラグランジュ乗数の値は 1.000 である。(また、どのような最適化問題であっても、目的関数に

設定された制約式 obj のラグランジュ乗数の値はつねに 1 であるから議論しない。)次の VAR ブロック

に、決定変数(内生変数)、および目的関数の最適解がある。 

このモデルの解、および、それに基づいて計算された変数の値は、伊藤 1.1.2 の説明と合致する(表

2.2)。部品の品質の解 x.l(i)における、メーカーの限界収入 db(i)は、サプライヤーi の限界費用

theta(i)に等しい。take-it-or-leave-it offer を考えて、サプライヤーがメーカーから受け取る価格

w(i)が費用 c(i)に等しいという条件 pc(i)を課しているので、サプライヤーの効用はゼロとなる。し

たがって、このメーカーの効用最大化問題を総余剰最大化問題と呼んでも構わない。 

表 2.2: 効率的サプライヤー・モデルの数値解 

変数 プログラム中での変数名 数値解 

部品の品質 x(i) 6.250 

メーカーの収入 b(i) 2.500 

メーカーの費用 c(i) 1.250 

解(x.l(i))で評価した ib の一階の微係数 db(i) 0.200 

メーカーの支払い価格 w(i) 1.250 

メーカーの効用 Util 1.250 

 

2.1.3 非効率的サプライヤーのモデル (PS2_F_inf.gms) 

非効率的サプライヤーのモデルの入力ファイル(PS2_F_inf.gms)をみていこう。問題としては、

(2.1)–(2.3)の最大化問題を infi = のみについて解けばよい。そのために非効率的サプライヤーのコ

ンピュータ・プログラムは、効率的サプライヤーのプログラム(リスト 2.1)の 2 箇所を変更するだけでよい。 

・8 行目の指示子 set で定義しているタイプ集合 i の要素を「inf」に書き替える。 

・11 行目の theta(i)の値を、非効率的サプライヤーの限界費用としてここで仮定する「0.3」に書き

替える。 

このモデルの解、および、それに基づいて計算された値は表 2.3 の通りである。部品の品質の解 

x.l(i)における、メーカーの限界収入 db(i)がサプライヤーの限界費用 theta(i)に一致すること

や、メーカーの効用最大化問題として定式化されたこの問題が実質的に総余剰最大化問題であること

などは、2.1.2 で指摘したことと同じである。 

 

表 2.3: 非効率的サプライヤー・モデルの数値解 

変数 プログラム中での変数名 数値解 

部品の品質 x(i) 2.778 

メーカーの収入 b(i) 1.667 

メーカーの費用 c(i) 0.833 

解(x.l(i))で評価した ib の一階の微係数 db(i) 0.300 

メーカーの支払い価格 w(i) 0.833 

メーカーの効用 Util 0.833 

2.1.4 両サプライヤーを統合したファーストベスト・モデル (PS2_F.gms) 

これまでに構築してきた 2 つのモデルを基に、効率的サプライヤーと非効率的サプライヤーの 2 種

類が、ある確率 ip で存在し、かつ、彼がどちらのタイプであるかがメーカーにとっても明らかであるとす

る状況の下で達成できるファーストベストの最適契約を考える。すなわち、問題(2.1)–(2.3)とほぼ同じも

のを両方のタイプのサプライヤー( infeffi ,= )が現れることを前提にして解けばよい。 

 ( )[ ]∑ −=
i

iii wxbpUtilmax      (2.1’) 

subject to 

 ( ) 5.0

ii xxb =  i∀       (2.2) 

 ruxw iii ≥−θ  i∀       (2.3’) 

前項のサプライヤーのタイプ別モデルとは目的関数(2.1’)が異なっている。ここではタイプ iの事前確率

ip で加重したメーカーの効用を最大化するようになっている。また、これまでのモデルでは、take-it-or-

leave-it offer を先験的に仮定して iw を ( )ii xc に置き換えて問題を簡単化していたが、ここではそのよ

うなことは仮定せずにそのままサプライヤー iへの支払い価格 iw を用いる。なお、支払い価格 iw がサ

プライヤーの効用 iixθ と留保効用 ru の和を上回る可能性を考えて(この点はつぎのセカンドベストの状

況で重要となる)、(2.3’)では不等号を用いる。 

 コンピュータ・プログラム上で、前項のものと異なる点を中心に説明する(リスト 2.3)。8 行目で、タイプを

表す添え字の集合 i の要素として効率的サプライヤーeff と非効率的サプライヤーinf の両方を入

れる。11–16 行目で、それぞれのタイプのサプライヤーの限界費用 theta(i)の値と、サプライヤーが

効率的 eff あるいは非効率的 inf である確率 p(i)を仮定する。17 行目で留保効用 ru を与える。

この数値モデルでは、伊藤にならってゼロに設定しているが、必要に応じて、ゼロ以外の適当な数値を

与えることができる。(前項のモデルでも同様。) 

 20–25 行目では、この最大化問題の決定変数を定義する。前項とは違って、費用 c(i)ではなく、メ

ーカーの支払い価格 w(i)を決定変数とする。26–29 行目は、目的関数と制約式の名前を定義してい

る。32–34 行目が最大化問題を構成する一連の制約式である。32 行目が目的関数 obj である。メー

カーの効用 Util は、各サプライヤーの作る部品から得られる利益(b(i)と w(i)の差額)に、効率的

サプライヤーと非効率サプライヤーが出現する確率 p(i)を掛けて合算したものである。33 行目は、そ

れぞれのサプライヤーからの収入関数 b(i)の定義式 rev(i)であり、前項で解説したコンピュータ・

モデルと同じ関数
5.0

ii xb = を仮定する。34 行目は、参加制約条件 pc(i)である。不等号(≥ )は、プロ

グラム中では「=g=」と表す。40–41 行目ではモデル名を定義して、メーカーの期待効用 Util を最大

化するようにモデルを解くことを指示している。 

 

リスト 2.3: 両サプライヤーを統合したファーストベスト・モデルの入力ファイル (PS2_F.gms) 

...  

7 * Definition of Set 

8 Set     i       type of supplier        /eff, inf/; 

9  

10 * Definition of Parameters 

11 Parameter 

12         theta(i)        efficiency       /eff    0.2 

13                                             inf     0.3/ 

14         p(i)             probability of type 

15                                             /eff    0.2 

16                                             inf     0.8/; 

17 Scalar  ru  reservation utility         /0/; 

18  

19 * Definition of Primal/Dual Variables 

20 Positive Variable 

21         x(i)            quality 

22         b(i)            maker's revenue 

23         w(i)            price; 

24 Variable 

25         Util            maker’s utility; 

26 Equation 

27         obj             supplier's profit function 

28         rev(i)          maker's revenue function 

29         pc(i)           participation constraint; 

30  

31 * Specification of Equations 

32 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

33 rev(i)..b(i) =e= x(i)**(0.5); 

34 pc(i).. w(i)-theta(i)*x(i) =e= ru; 

35  

36 * Setting Lower Bounds on Variables to Avoid Division by Zero 

37 x.lo(i)=0.0001;; 

38  

39 * Defining and Solving the Model 

40 Model FB1 /all/; 

41 Solve FB1 maximizing Util using NLP; 

42  

43 * End of Model 

 

 以上が入力ファイルである。 続いて、この出力ファイル(PS2_F.lst)を確認する(リスト 2.4)。まず、

EQU のブロックでは、制約式 pc(i)のラグランジュ乗数(128–133 行目の MARGINAL 値)が、効率的

サプライヤーeff、非効率的サプライヤーinf ともに、ゼロではないので、この制約が有効であることが

分かる。すなわち制約は等号で成立しており(メーカーからサプライヤー iへの支払い価格 iw がそれぞ

れの費用 iixθ に厳密に等しい)、したがってサプライヤーの効用水準がゼロである(正確には留保効用

水準に一致する)ことがわかる。VAR ブロックでは、決定変数である品質 x(i)、価格 w(i)、メーカーの

収入 b(i)、およびメーカーの効用 Util の解が示されている。 

 

リスト 2.4: 両サプライヤーを統合したファーストベスト・モデルの出力ファイル(抜粋) (PS2_F.lst) 

...   

74                S O L V E      S U M M A R Y 

75  

76      MODEL   FB1                 OBJECTIVE  Util 

77      TYPE    NLP                 DIRECTION  MAXIMIZE 

78      SOLVER  CONOPT              FROM LINE  41 

79  

80 **** SOLVER STATUS     1 NORMAL COMPLETION          

81 **** MODEL STATUS      2 LOCALLY OPTIMAL            

82 **** OBJECTIVE VALUE                0.9167 

...   

115                        LOWER     LEVEL     UPPER    MARGINAL 

116  

117 ---- EQU obj             .         .         .        1.000       

118  

119   obj  supplier's profit function 

120  

121 ---- EQU rev  maker's revenue function 

122  

123        LOWER     LEVEL     UPPER    MARGINAL 

124  

125 eff      .         .         .        0.200       

126 inf      .         .         .        0.800       

127  

128 ---- EQU pc  participation constraint 

129  

130        LOWER     LEVEL     UPPER    MARGINAL 

131  

132 eff      .         .         .       -0.200       

133 inf      .         .         .       -0.800       

134  

135 ---- VAR x  quality 

136  

137        LOWER     LEVEL     UPPER    MARGINAL 

138  

139 eff 1.0000E-7   6.250     +INF  2.0951E-9       

140 inf 1.0000E-7   2.778     +INF       EPS        

141  

142 ---- VAR b  maker's revenue 

143  

144        LOWER     LEVEL     UPPER    MARGINAL 

145  

146 eff      .        2.500     +INF       .          

147 inf      .        1.667     +INF       .          

148  

149 ---- VAR w  price 

150  

151        LOWER     LEVEL     UPPER    MARGINAL 

152  

153 eff      .        1.250     +INF       .          

154 inf      .        0.833     +INF       .          

155  

156                        LOWER     LEVEL     UPPER    MARGINAL 

157  

158 ---- VAR Util        -INF      0.917     +INF       .          

...   

 

このモデルの解を表 2.4 にまとめた。メーカーの効用以外はどの値も、前項で効率的サプライヤーと

非効率的サプライヤーの問題を別々にモデル化したときの解(表 2.2–2.3)と一致している。 

表 2.4: 両サプライヤーを統合したファーストベスト・モデルの数値解 

変数 プログラム中での変数名 サプライヤーのタイプ 数値解 

部品の品質 x(i) 効率的(eff) 6.250 

  非効率的(inf) 2.778 

メーカーの収入 b(i) 効率的(eff) 2.500 

  非効率的(inf) 1.667 
w(i) 効率的(eff) 1.250 メーカーの支払い価格 
 非効率的(inf) 0.833 

メーカーの効用 Util  0.917 

 

2.1.5 両サプライヤーを統合したセカンドベスト・モデル (PS2_S.gms)  

 部品調達問題のセカンドベスト・モデルは、メーカーが、サプライヤーのタイプ、すなわち、サプライヤ

ーの提供する部品製造の限界費用 iθ を知らないままに、部品の品質 ix とサプライヤーに支払うべき価

格 iw を含む契約を作るものである。効率的サプライヤーが限界費用を真実のものより高く申告すること

で(すなわち非効率的サプライヤーの振りをして)、その虚偽の限界費用に応じて過大に支払われる価

格と真実の費用との間の差額を懐に入れる可能性がある。メーカーはそれを防ぐために、タイプを偽っ

て申告してもトクにならない条件—誘因両立制約—をあらたに考慮して最適契約を考えることになる。 

 セカンドベスト・モデルは、伊藤 1.1.4 の問題( p )にそのまま従う。これは、本稿 2.1.4 のファーストベ

ストの問題(2.1’), (2.2), (2.3’)に、誘因両立制約(2.4)を付加することで作ることができる。なお、ファース

トベスト・モデルは 2.1.2 や 2.1.3 のように効率的サプライヤーのモデルと非効率的サプライヤーのモデ

ルを別々に作ることができるが、セカンドベスト・モデルは両サプライヤーを統合したモデルでなければ

ならない。それは、誘因両立制約に、両方のサプライヤーの品質 x(i)とメーカーの支払い w(i)を、

決定変数として入れなければならないからである。 

 セカンドベスト･モデルは以下の通りである。 

 ( )[ ]∑ −=
i

iii wxbpUtilmax      (2.1’) 

subject to 

 ( ) 5.0

ii xxb =  i∀       (2.2) 

 ruxw iii ≥−θ  i∀       (2.3’) 

 jijiii xwxw θθ −≥−  ji ≠∀      (2.4) 

iθ の限界費用を持つタイプ iのサプライヤーにとって、(2.4)の左辺は自らのタイプを正直に申告したと

きに提示される価格 iw と品質 ix から得られる効用であり、その右辺は、自らのタイプを ( )ij ≠ と偽って

申告したときに提示される価格 jw と品質 jx から得られる効用である。左辺が右辺を下回らないようにす

ることで、自らのタイプを正直に申告させることができる。リスト 2.5 では、タイプの記号を iと同じ意味で

j も用いるために、9 行目で Alias 指示子を使って、j が i の別名(alias)であり、互換的に用いる旨

を宣言している。 

 この(タイプ i のサプライヤーが自らのタイプについて「タイプ j である」と申告することを防ぐ)誘因両

立制約(2.4)の名称 ic(i,j)は 30 行目で定義され、その制約式自体は 36 行目で記述される。ところ

で、 infeffi ,= についてプログラム中の制約式 ic(i,j)に相当するものの内容を書き下してみると、 

 infeffinfeffeffeff xwxw θθ −≥−      (2.4a) 

 effeffeffeffeffeff xwxw θθ −≥−      (2.4b) 

 effinfeffinfinfinf xwxw θθ −≥−      (2.4c) 

 infinfinfinfinfinf xwxw θθ −≥−      (2.4d) 

の 4 本の制約を課していることがわかる。すなわち、このプログラム中の制約式は、本来 2 本しかない

制約式(2.4)に比べて、(2.4b)と(2.4d)の 2 本の冗長な制約式を含んでいる。しかしながら、これら 2 本

の冗長な制約式は、一見してすぐにわかるように右辺と左辺が同じものであるから常に等号で成立する

自明なものである。したがって、解いている問題に本質的な違いはない。10プログラムの他の部分は、両

サプライヤーを統合した前項のファーストベスト・モデル(PS2_F.gms)と変わらない。 

                                                      

10 もしこの冗長さが気になるのであれば、プログラム中の制約式を、 

 ic(i,j)$(ord(i) ne ord(j))..w(i)-theta(i)*x(i) =g= w(j)-theta(i)*x(j); 

 

と書き換えればよい。ここで「$(…)」は括弧内を条件とするダミー変数であり、「ord(i)」は添え字 i が、それが定

義された集合の中で何番目に現れるのか、その順番を表す関数であり、「ne」は≠ の意味である。 

リスト 2.5: セカンドベスト・モデルの入力ファイル (PS2_S.gms) 

...  

7 * Definition of Set 

8 Set     i       type of supplier        /eff, inf/; 

9 Alias (i,j); 

10 * Definition of Parameters 

11 Parameter 

12         theta(i)        efficiency       /eff    0.2 

13                                              inf     0.3/ 

14         p(i)             probability of type 

15                                              /eff    0.2 

16                                              inf     0.8/; 

17 Scalar  ru              reservation utility     /0/; 

18  

19 * Definition of Primal/Dual Variables 

20 Positive Variable 

21         x(i)            quality 

22         b(i)            maker's revenue 

23         w(i)            price; 

24 Variable 

25         Util            maker's utility; 

26 Equation 

27         obj             maker’s utility function 

28         rev(i)          maker's revenue function 

29         pc(i)           participation constraint 

30         ic(i,j)         incentive compatibility constraint; 

31  

32 * Specification of Equations 

33 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

34 rev(i)..b(i) =e= x(i)**(0.5); 

35 pc(i).. w(i)-theta(i)*x(i) =g= ru; 

36 ic(i,j)..w(i)-theta(i)*x(i) =g= w(j)-theta(i)*x(j); 

37  

38 * Setting Lower Bounds on Variables to Avoid Division by Zero 

39 x.lo(i)=0.0001;; 

40  

41 * Defining and Solving the Model 

42 Model SB1 /all/; 

43 Solve SB1 maximizing Util using NLP; 

44  

45 * End of Model 

 

この入力ファイルを基に問題を解いた結果は、リスト 2.6 に示されている。 

リスト 2.6: セカンドベスト・モデルの出力ファイル(抜粋) (PS2_S.lst) 

...  

123 ---- EQU rev  maker's revenue function 

124  

125        LOWER     LEVEL     UPPER    MARGINAL 

126  

127 eff      .         .         .        0.200       

128 inf      .         .         .        0.800       

129  

130 ---- EQU pc  participation constraint 

131  

132        LOWER     LEVEL     UPPER    MARGINAL 

133  

134 eff      .        0.237     +INF       .          

135 inf      .         .        +INF     -1.000       

136  

137 ---- EQU ic  incentive compatibility constraint 

138  

139            LOWER     LEVEL     UPPER    MARGINAL 

140  

141 eff.inf      .         .        +INF     -0.200       

142 inf.eff      .        0.388     +INF       EPS        

143  

144 ---- VAR x  quality 

145  

146        LOWER     LEVEL     UPPER    MARGINAL 

147  

148 eff 1.0000E-7     6.250     +INF  6.7551E-8       

149 inf 1.0000E-7     2.367     +INF       .          

150  

151 ---- VAR b  maker's revenue 

152  

153        LOWER     LEVEL     UPPER    MARGINAL 

154  

155 eff      .        2.500     +INF       .          

156 inf      .        1.538     +INF       .          

157  

158 ---- VAR w  price 

159  

160        LOWER     LEVEL     UPPER    MARGINAL 

161  

162 eff      .        1.487     +INF       .          

163 inf      .        0.710     +INF       .          

164  

165                        LOWER     LEVEL     UPPER    MARGINAL 

166  

167 ---- VAR Util           -INF      0.865     +INF       .          

GRIPS Policy Research Center Discussion Paper : 10-34



H. Hashimoto, K. Hamada, & N. Hosoe 

Contract Theory: A Programming-Model Approach  Page 22 

...  

 

2.1.6 ファーストベスト・モデルとセカンドベスト・モデルの解の比較 

 ファーストベストとセカンドベストを比較した表 2.5 の結果は次のようにまとめられ、伊藤 1.1.4 の解説と

合致することはこれまでと同様である。すなわち、誘因両立制約は、効率的サプライヤーeff について

だけが有効である(この制約のラグランジュ乗数がゼロでない)。すなわち、自らのタイプを偽ったときと、

正直に申告したときの間で無差別となっている。メーカーは、効率的サプライヤーに虚偽の申告をさせ

ないようにするために、情報の非対称性がなかったときと比べて、より高い価格 effw を「効率的」と申告し

たものに提示する。その一方で、(単純に、「効率的」と申告したものへの提示価格 effw を引き上げるだ

けでは、効率的サプライヤーが得る情報レントが大きくなりすぎるので、 effw の infw に対する相対的な

価格 infeff ww / を引き上げるように)、より低い価格 infw を「非効率的」と申告したものに提示している。

すなわち、サプライヤーへの支払い価格 w(i)は、セカンドベストでは、ファーストベストに比べ、効率的

サプライヤーeff で増加し、非効率的サプライヤーinf で減少する。この結果、効率的サプライヤーに

は、参加制約 pc(i)の両辺の差額(スラック, slack)0.237—これはリスト 2.6 の 134 行目、EQU ブロッ

クにある pc(“eff”)の LEVEL 値と LOWER 値の差として表される—に等しい情報レントが発生してい

る。11現在のモデルでは留保効用をゼロと仮定しているので、情報レントが効率的サプライヤーの効用と

なる。 

 

                                                      

11 EQU ブロックの LOWER, LEVEL, UPPER 値それぞれが何を示しているかについては McCarl(2009, 第 2.4.5.2

項)参照。なお、Solve 命令の前(のどこでもよい)に「Option solslack=1;」というオプションを記述しておくと、

出力ファイルの EQU ブロックには、LEVEL の代わりに SLACK がそのまま出力される。 

表 2.5: ファーストベスト・モデルとセカンドベスト・モデルの解の比較 

変数・制約 プログラム中での

変数名 
サプライヤーのタイ

プ 
セカンドベスト・ 
モデルの解 

ファーストベスト・ 
モデルの解* 

部品の品質 x(i) 効率的(eff) 6.250 6.250 

  非効率的(inf) 2.367 2.778 

     
b(i) 効率的(eff) 2.500 2.500 メーカーの収入 
 非効率的(inf) 1.538 1.667 

     
w(i) 効率的(eff) 1.487 1.250 メ ー カ ー の 支 払

い価格  非効率的(inf) 0.710 0.833 

     
メーカーの効用 Util  0.865 0.917 
     

効率的(eff) 0.237 0.000 サプライヤーの効

用(=情報レント) 

pc.l(i) 
-pc.lo(i) 非効率的(inf) 0.000 0.000 

    
ic(”eff”,”inf”) 

効率的が非効率的を偽る 
–0.200 – 誘因両立制約の

ラグランジュ乗数 
ic(”inf”,”eff”) 

非効率的が効率的を偽る 
0.000 – 

     
pc.m(i) 効率的(eff) 0.000 –0.200 参加制約のラグラ

ンジュ乗数  非効率的(inf) –0.100 –0.800 

注: ファーストベスト・モデルの数値解の一部は、表 2.4 に示したものと同じものである。 

 

 一方、非効率的サプライヤーには、誘因両立制約が有効ではない(=不等号で成立)ことからわかるよう

に、虚偽の申告しても効用が増加しないために、自らのタイプを偽る誘因はない。その結果、非効率的

サプライヤーの参加制約が有効になり、すべての情報レントはメーカーに吸い上げられてしまって、得ら

れる効用は留保効用水準(これはゼロと仮定されている)だけにとどまる。なお、ファーストベストにおいて

は、両サプライヤーについて自らのタイプに関する情報はメーカーに完全に見透かされているから情報

レントは発生せず、どちらの参加制約も有効であった。 

 品質 x(i)については、効率的サプライヤーeff のセカンドベスト解は、ファーストベスト解に等しい。

これに対して、非効率的サプライヤーinf のセカンドベスト解は、上で議論したように、効率的サプライ

ヤーに虚偽の申告をさせないように非効率的サプライヤーに対する提示価格を引き下げたために、ファ

ーストベスト解に比べ過小になる。また、メーカーの期待効用 Util は、ファーストベストに比べて、セカ

ンドベストでは減少する。この減少は、効率的サプライヤーが留保効用水準を上回る効用水準を情報レ

ントとして 0.237 だけ得るようになったことと、非効率的サプライヤーへの提示価格が低下して品質が非

効率的になったこと(上の数値問題では infx が 0.411 だけ低下した)のためである。 

 

2.1.7 ファーストベストとセカンドベストを同時に解く (PS2_F&S.gms) 

 これまでは、2 つの入力ファイルを作成し、それぞれの中でファーストベスト・モデル(PS2_F.gms)とセ

カンドベスト・モデル(PS2_S.gms)を別々に解いていた。両者を比較するとこれらの違いは誘因両立制

約 ic(i,j)の有無のみである。そこで、1 つのプログラムの中で、これら 2 つの問題を同時に解くプロ

グラムを紹介する(リスト 2.7)。プログラムの 39 行目まではセカンドベスト・モデル(リスト 2.5)とまったく同

じである。42 行目で obj, rev(i), pc(i)の 3 つの制約式(目的関数となるべきものも含む、以下同

じ)から構成されるファーストベスト・モデルを FB2 と名付け、これを 43 行目で Util を最大化するよう

にしている。45 行目では、obj, rev(i), pc(i)、および、ic(i,j)の 4 つの制約式から構成される

セカンドベスト・モデルを SB2 と名付け、これを 46 行目で最大化問題として解くようにしている。 

 このモデルに関してプログラム技術上で注意するべき点は、2 点ある。これまでのプログラムでは、

Model 指示子でモデル名を定義する際には、(その行よりも前の行で)定義したすべての制約式を用い

てモデルを構成することを考えていたので、 

 Model モデル名  /all/; 

 

と記述していた。しかし、より柔軟に、どの制約式を用いてモデルを構築するかを選択することもできる。

その際には、 

 Model モデル名  /制約式1, 制約式2, …/; 

 

というように、モデルに含むべき制約式を(コンマで区切って)列挙する。ただし、制約式に添え字がつい

ていても、そこでは添え字をつけない。 

リ ス ト 2.7: フ ァ ー ス ト ベ ス ト ・ モ デ ル と セ カ ン ド ベ ス ト ・ モ デ ル を 同 時 に 解 く 入 力 フ ァ イ ル 

(PS2_F&S.gms) 

...  

7 * Definition of Set 

8 Set     i       type of supplier        /eff, inf/; 

9 Alias (i,j); 

10 * Definition of Parameters 

11 Parameter 

12         theta(i)        efficiency      /eff    0.2 

13                                             inf     0.3/ 

14         p(i)             probability of type 

15                                             /eff    0.2 

16                                             inf     0.8/; 

17 Scalar  ru              reservation utility     /0/; 

18  

19 * Definition of Primal/Dual Variables 

20 Positive Variable 

21         x(i)            quality 

22         b(i)            maker's revenue 

23         w(i)            price; 

24 Variable 

25         Util            maker's utility; 

26 Equation 

27         obj             maker’s utility function 

28         rev(i)          maker's revenue function 

29         pc(i)           participation constraint 

30         ic(i,j)         incentive compatibility constraint; 

31  

32 * Specification of Equations 

33 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

34 rev(i)..b(i) =e= x(i)**(0.5); 

35 pc(i).. w(i)-theta(i)*x(i) =g= ru; 

36 ic(i,j)..w(i)-theta(i)*x(i) =g= w(j)-theta(i)*x(j); 

37  

38 * Setting Lower Bounds on Variables to Avoid Division by Zero 

39 x.lo(i)=0.0001;; 

40  

41 * Defining and Solving the Model 

42 Model FB1 /obj,rev,pc/; 

43 Solve FB1 maximizing Util using NLP; 

44  

45 Model SB1 /obj,rev,pc,ic/; 

46 Solve SB1 maximizing Util using NLP; 

47  

48 * End of Model 

 いまひとつは、1 つの入力ファイルの中で複数の Solve 命令を書くことで複数のモデルを解くことが

でき、そのとき、出力ファイルには Solve 命令の順番通りに同じ数だけの SOLVE SUMMARY が出力さ

れることである。このモデルの出力ファイル(PS2_S.lst)では、それぞれの SOLVE SUMMARY の冒頭

にモデル名(81 行目に FB1、197 行目に SB1)が示される(リスト 2.8)。これらの解が、ファーストベスト・

モデル(PS2_F.gms)とセカンドベスト・モデル(PS2_S.gms)を別々に解いた結果(表 2.5)と一致するこ

とも確認できるであろう。 

 

リスト 2.8: ファーストベスト・モデルとセカンドベスト・モデルを同時に解いた出力ファイル(抜粋) 

(PS2_F&S.lst) 

...  

79                S O L V E      S U M M A R Y 

80  

81      MODEL   FB1                 OBJECTIVE  Util 

82      TYPE    NLP                 DIRECTION  MAXIMIZE 

83      SOLVER  CONOPT              FROM LINE  43 

84  

85 **** SOLVER STATUS     1 NORMAL COMPLETION          

86 **** MODEL STATUS      2 LOCALLY OPTIMAL            

87 **** OBJECTIVE VALUE                0.9167 

...  

195                S O L V E      S U M M A R Y 

196  

197      MODEL   SB1                 OBJECTIVE  Util 

198      TYPE    NLP                 DIRECTION  MAXIMIZE 

199      SOLVER  CONOPT              FROM LINE  46 

200  

201 **** SOLVER STATUS     1 NORMAL COMPLETION          

202 **** MODEL STATUS      2 LOCALLY OPTIMAL            

203 **** OBJECTIVE VALUE                0.8654 

...  

 

2.2 3 タイプ・モデル 

2.2.1 モデルの概要と 2 タイプ・モデルとの相違点 

 前節では、伊藤 1.1「例 部品調達の問題」に従い、効率的と非効率的の 2 タイプ・モデルを作った。

これに対して、ここでは伊藤 1.3「分析 タイプが離散変数のケース」の応用として、3 タイプ(0, 1, 2 であ

り、小さい数字であるほど効率的)をもつモデルを作る。 

 2 タイプ・モデルと N タイプ・モデルの間で、大きく異なってくるのは、セカンドベスト・モデルにおける

誘因両立制約である。2 タイプ・モデルでは、eff が inf と偽ることと、inf が eff と偽ることの 2 つ

を防止する条件を課せばすべての場合を網羅できたが、タイプが N 個の場合は、同様に「総当たり」 

すなわち、各サプライヤーにとって自分以外の全てのサプライヤーとの組み合わせを考えると、全部で

( )NN 1− 本の誘因両立制約が必要になる。たとえば、3 タイプであれば 6 本で済むが、10 タイプであ

れば 90 本にものぼる。 

この「総当たり」に対応した誘因両立制約を大域的条件(または大域的誘因両立制約)と呼んでいる。

2 タイプ･モデルではあるが、この制約を課したものが伊藤 1.1.4 の問題( p )である。これに対して、伊

藤 1.3 では、追加的な条件(Spence-Mirrlees の単一交差性、以下 SCP)を仮定して、大域的な誘因

両立制約を、より簡単な条件—これを局所的条件(または局所的誘因両立制約と呼ぶ)—に置き換えて

問題を簡単化して解いている。(SCP については、セカンドベスト・モデルのところで詳しく説明する。)こ

うして作られたものが、伊藤 1.3.2 の問題( NP )である。さらにこの局所的条件も、解の単調性を仮定す

ることにより、本来 2 本で構成される局所的条件のうちの 1 本のみを考えることで済む。これに従い、次

の問題( NP' )では、単調性制約(MN)を外して解いた上で、解が単調性を満たすことを示すという方法を

取っている。本稿ではまずこの簡単化した問題を解くことにする。 

こうした伊藤の議論に従い、2.2.2 でファーストベスト・ベストモデル(ここでは誘因両立制約が必要な

いので、SCP や単調性は無関係)を作った後で、2.2.3 ではそのファーストベスト・モデルを基に、SCP

と単調性が成立するとして、局所的条件を使ったセカンドベスト・モデルを、次いで、2.2.4 では同じモデ

ルで、局所的条件を大域的条件に置き換えたものを解き、両者が同一の解を得ることを示す。SCP が

成立していないモデルや単調性が成立していないモデルは、次節のモデルの拡張のところで説明する。 

なお、伊藤 1.1 と 1.3 の間で記号表現や関数の定義が異なっている点がある。本稿では伊藤 1.1 の

表現に従うものとしているので、伊藤 1.3 と 2 つの点で表現が異なる。第 1 は記号表現上の違いである。

伊藤 1.3 では、限界費用を描写するパラメータ iθ のタイプ iに関する順序づけが、添え字の数字が大き

いほど効率的なサプライヤーということに変わっている(表 2.1)。しかし本稿では一貫して伊藤 1.1 に従

い、タイプ 0 が最も効率的であるとする。この結果いくつかの部分で伊藤 1.3 の記述を読み替える必要

があり、適宜指摘する。第 2 は目的関数の表現上の違いである。伊藤 1.3 では、最終的に目的関数や

制約式を、総余剰 S やサプライヤーの効用U を用いて表現しなおした問題(
''

NP )を用いて議論してい

る。しかし、本稿ではこれまで通りに、 ( )[ ]∑ −
i iii wxbp を最大化し、制約条件もこれまで通り ix と iw

の関係として記述する。 

 

 

2.2.2 ファーストベスト・モデル (PS3_F.gms) 

3 タイプのファーストベスト･モデルは、本稿 2.1.4 の 2 タイプのファーストベスト・モデルを単純に拡張

すればよい(リスト 2.9)。8 行目で 3 つのタイプを定義している点、12–14 行目で theta(i)をタイプ 0

が最も効率的(限界費用が低い)になるように定義・設定している。そのほか、モデルの名前を 42 行目

で(FB1 ではなく)FB2 とした点が異なっている。 

 

リスト 2.9: 3 タイプのファーストベスト・モデルのための入力ファイル (PS3_F.gms) 

...  

7 * Definition of Set 

8 Set     i       type of supplier        /0,1,2/; 

9 Alias (i,j); 

10 * Definition of Parameters 

11 Parameter 

12         theta(i)        efficiency      /0      0.1 

13                                             1       0.2 

14                                             2       0.3/ 

15         p(i)             probability of type 

16                                             /0      0.2 

17                                             1       0.5 

18                                             2       0.3/; 

19 Scalar  ru              reservation utility     /0/; 

20  

21 * Definition of Primal/Dual Variables 

22 Positive Variable 

23         x(i)            quality 

24         b(i)            maker's revenue 

25         w(i)            price; 

26 Variable 

27         Util            maker's utility; 

28 Equation 

29         obj             maker’s utility function 

30         rev(i)          maker's revenue function 

31         pc(i)           participation constraint 

32  

33 * Specification of Equations 

34 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

35 rev(i)..b(i) =e= x(i)**(0.5); 

36 pc(i).. w(i)-theta(i)*x(i) =g= ru; 

37  

38 * Setting Lower Bounds on Variables to Avoid Division by Zero 

39 x.lo(i)=0.0001;; 

40  

41 * Defining and Solving the Model 

42 Model FB2 /all/; 

43 Solve FB2 maximizing Util using NLP; 

44  

45 * End of Model 

 

2.2.3 局所的誘因両立制約を用いたセカンドベスト・モデル (PS3_S.gms) 

タイプが 2 個のモデルから 3 個のモデルに拡張されることで複雑になる(大局的)誘因両立制約を、

SCP の仮定を用いて、より簡単な局所的条件に置き換えたり、単調性を仮定したりして、モデルを簡単

化できることは先に述べたとおりである。本項のセカンドベスト・モデルと前項のファーストベスト・モデル

との違いは、局所的誘因両立制約が加わっている 1 点のみである。 

局所的条件の詳細な説明に入る前に、以下では、 N タイプ・モデルの誘因両立制約を一般的に検

討することにする。一般にタイプ iが自らのタイプをどのように表明するかを考えると、表 2.6 の薄い灰色

と濃い灰色の部分すべてについて可能性があるので、大域的誘因両立制約を考えなければならない。 

表 2.6: 大域的誘因両立制約と局所的誘因両立制約 

表明するタイプ  

0 … 1−i  i  1+i  … 1−N  

0 ( )00 |θθU  … ( )01 |θθ −iU  ( )0|θθiU  ( )01 |θθ +iU  … ( )01 |θθ −NU  

…    …    

1−i     ( )1| −iiU θθ     

i  ( )iU θθ |0  … ( )iiU θθ |1−  ( )iiU θθ |  ( )iiU θθ |1+  … ( )iNU θθ |1−  

1+i     ( )1| +iiU θθ     

…    …    

真
の

タ
イ

プ
 

1−N     ( )1| −NiU θθ     

 

 このとき、タイプ iに正直に自らのタイプを申告させるための条件は(タイプ iがタイプ jであると申告し

たときの効用を ( )ijU θθ | として)、 

 ( ) ( )ijii UU θθθθ || ≥  j∀  

となる。書き下すと、大域的誘因両立制約は、 

 ( ) ( )iii UU θθθθ || 0≥  

 ( ) ( )iii UU θθθθ || 1≥  

 ( ) ( )iii UU θθθθ || 2≥  

 … 

 ( ) ( )iiii UU θθθθ || 1−≥  

 ( ) ( )iiii UU θθθθ || ≥  (だだしこれは自明な条件) 

 ( ) ( )iiii UU θθθθ || 1+≥  

 … 

 ( ) ( )iNii UU θθθθ || 1−≥  

となり数が極端に多くなる。 

そこで、効用関数に一定の仮定をおいてこの誘因両立制約を簡素化することを考える。具体的には、

SCP を満たす場合を考える。(ただし、パラメータ iθ に関する順序づけが伊藤 1.3.2 と本稿との間で反

対になっているので、 xu が iθ の厳密な減少関数であることがこの条件である。)12この数値問題では、サ

プライヤー iの効用 ( )iiii xuwU θ,+= を構成する関数uについて、 ( ) iiii xxu θθ −=, と特定化して

いるので、関数uの ix に関する一階微分 xu は iθ に対して厳密に減少関数となっている。したがって、

実際、追加的な仮定なしにそのままで SCP を満たしているので、大域的条件の代わりに、局所的誘因

両立制約を使って問題を簡単化することができる。 

伊藤 1.3.2 では、この大域的条件を使ったセカンドベスト・モデルを、2 つの点で簡単化した上で問

題を解いている。第 1 は伊藤 1.3.2 の問題( NP )であり、LICD と LICU の両方の制約を置くものであ

る。すなわち、表 2.6 における局所的条件(濃い灰色)、 

 ( ) ( )iiii UU θθθθ || 1−≥  i∀      (LICU) 

 ( ) ( )iiii UU θθθθ || 1+≥  i∀      (LICD) 

だけを課すものである。13 

第 2 は、伊藤 1.3.2 の問題( NP' )であり、LICD と LICU のうちの 1 つの代わりとして、品質 ix の解

の単調性の条件(MN)、 

 Nxx ≥≥ ...0  

を置くものである。14LICD を例に取ると、それは各サプライヤーが効率性の点ですぐ 1 つ劣るタイプで

あると偽って申告するのを防ぐ条件だけを課すというものである。さらに手続き的には、単調性制約を無

視して問題を解いた後に、得られた解が単調性を満たしていることを示すことにより、単調性制約(MN)

                                                      

12 すなわち、 Xx∈∀ , Θ∈∀ ',θθ に対して ( ) ( )',,' θθθθ xuxu xx <⇒> が成り立つ場合。 

13 パラメータθ に関する順序づけが伊藤 1.3.2 と本稿との間で反対になっているので、より効率的なタイプであると

偽って申告させないための伊藤の条件 LICU 中に現れる ( )iiU θθ |1+ は、本稿では ( )iiU θθ |1− に置き換えら

れているが本質的に同じものである。LICD についても同様である。 

14 パラメータθ に関する順序づけが反対になっているので、不等号の向きが伊藤 1.3.2 の表記と逆になっている

が本質的には何も違わない。 

を無視することを正当化している。逆に単調性が満たされていないときは、LICD のほか、単調性制約

か LICU かのいずれかを必ず含んでいなければならない。単調性については、伊藤 1.3.2 の仮定 1.4

に関連して、2.3.2 で詳しく説明する。 

以下では実際に、前項で提示したファーストベスト･モデルを基にして、局所的条件 LICD を持つが、

単調性(MN)を外した形でセカンドベスト・モデルを解き、その後に解がこの単調性を満たしていることを

確認する(PS3_S.gms)。 

 

 ( )[ ]∑ −=
i

iii wxbpUtilmax      (2.1’) 

subject to 

 ( ) 5.0

ii xxb =  i∀       (2.2) 

 ruxw iii ≥−θ  i∀       (2.3’) 

 11 ++ −≥− iiiiii xwxw θθ  i∀      (2.5) 

 なお、伊藤の問題( NP' )では、参加制約(2.3’)を最も非効率なサプライヤー(本稿の表記では 2=i )

のみについて課しているが、本稿では前項のファーストベスト･モデルと同様にすべてのサプライヤーに

ついてこれを課す(リスト 2.10 の 37 行目)。もちろん、伊藤が示すとおり、 2=i 以外についてはこの制

約式は無効であるから、解がゆがむおそれはない。一方、2.1.6 で触れたように、この制約式のスラック

(左辺と右辺の差)を見ることで各サプライヤーが得る情報レントを確認することができる。LICD(2.5)は、

入力ファイルの 38 行目に現れる。サプライヤーi から見て 1 つ効率性の点で劣る(添え字 i の順番で

は 1 つ後の)サプライヤーを表す添え字は、プログラム中では直感通り i+1 とすればよい。 

 

リスト 2.10: 3 タイプのセカンドベスト・モデル (PS3_S.gms) 

...  

7 * Definition of Set 

8 Set     i       type of supplier        /0,1,2/; 

9 Alias (i,j); 

10 * Definition of Parameters 

11 Parameter 

12         Theta(i)        efficiency       /0      0.1 

13                                              1       0.2 

14                                              2       0.3/ 

15         p(i)             probability of type 

16                                              /0      0.2 

17                                              1       0.5 

18                                              2       0.3/; 

19 Scalar  ru              reservation utility     /0/; 

20  

21 * Definition of Primal/Dual Variables 

22 Positive Variable 

23         x(i)            quality 

24         b(i)            maker's revenue 

25         w(i)            price; 

26 Variable 

27         Util            maker's utility; 

28 Equation 

29         obj             maker’s utility function 

30         rev(i)          maker's revenue function 

31         pc(i)           participation constraint 

32         licd(i)        incentive compatibility constraint; 

33  

34 * Specification of Equations 

35 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

36 rev(i)..b(i) =e= x(i)**(0.5); 

37 pc(i).. w(i)-theta(i)*x(i) =g= ru; 

38 licd(i)..w(i)-theta(i)*x(i) =g= w(i+1)-theta(i)*x(i+1); 

39  

40 * Setting Lower Bounds on Variables to Avoid Division by Zero 

41 x.lo(i)=0.0001;; 

42  

43 * Defining and Solving the Model 

44 Model SB3 /all/; 

45 Solve SB3 maximizing Util using NLP; 

46  

47 * End of Model 

 

 この結果は、リスト 2.11 に示す出力ファイルのとおりである。 

 

リスト 2.11: 3 タイプのセカンドベスト・モデルの出力ファイル(抜粋) (PS3_S.lst) 

...  

125 ---- EQU rev  maker's revenue function 

126  

127      LOWER     LEVEL     UPPER    MARGINAL 

128  

129 0      .         .         .        0.200       

130 1      .         .         .        0.500       

131 2      .         .         .        0.300       

132  

133 ---- EQU pc  participation constraint 

134  

135      LOWER     LEVEL     UPPER    MARGINAL 

136  

137 0      .        0.522     +INF       .          

138 1      .        0.088     +INF       .          

139 2      .         .        +INF     -1.000       

140  

141 ---- EQU licd  incentive compatibility constraint 

142  

143      LOWER     LEVEL     UPPER    MARGINAL 

144  

145 0      .         .        +INF     -0.200       

146 1      .         .        +INF     -0.700       

147 2      .         .        +INF       .          

148  

149 ---- VAR x  quality 

150  

151      LOWER     LEVEL     UPPER    MARGINAL 

152  

153 0 1.0000E-7    25.000     +INF  5.7724E-8       

154 1 1.0000E-7     4.340     +INF       EPS        

155 2 1.0000E-7     0.879     +INF       EPS        

156  

157 ---- VAR b  maker's revenue 

158  

159      LOWER     LEVEL     UPPER    MARGINAL 

160  

161 0      .        5.000     +INF       .          

162 1      .        2.083     +INF       .          

163 2      .        0.937     +INF       .          

164  

165 ---- VAR w  price 

166  

167      LOWER     LEVEL     UPPER    MARGINAL 

168  

169 0      .        3.022     +INF       .          

170 1      .        0.956     +INF       .          

171 2      .        0.264     +INF       .          

172  

173                        LOWER     LEVEL     UPPER    MARGINAL 

174  

175 ---- VAR Util           -INF      1.161     +INF       .          

...  

 

2.2.4 ファーストベスト・モデルとセカンドベスト･モデルの解の比較 

 2 タイプ・モデルの結果について 2.1.6 で比較したときと同様に、ファーストベスト解と比較してセカンド

ベスト解がどのように変化したか、伊藤 1.3.2 の説明に照らしながら検討する(表 2.7)。なお、先にも述

べたようにこのセカンドベスト・モデルは SCP を満たしているので、局所的制約モデルの解と後に述べ

る大域的制約モデルの解は等しい。 

 

表 2.7: 3 タイプ・モデルのファーストベストとセカンドベストの比較 

変数 プロ グ ラ ム 中

での変数名 
サプライヤ

ーのタイプ 
セカンドベスト・

モデルの解 
ファーストベス

ト・モデルの解 
両者の差 

部品の品質 x(i) 0 25.00 25.00 0.00 

  1 4.34 6.25 –1.91 
  2 0.88 2.78 –1.90 
      
メーカーの収入 b(i) 0 5.00 5.00 –0.00 
  1 2.08 2.50 –0.42 
  2 0.94 1.67 –0.73 
      

w(i) 0 3.02 2.50 0.52 
 1 0.96 1.25 –0.29 

メーカーの支払

い価格 
  2 0.26 0.83 –0.57 

      
メーカーの効用 Util  1.16 1.37 –0.21 
      

0 0.52 0.00 0.52 pc.l(i) 
-pc.lo(i) 1 0.09 0.00 0.09 

サプライヤーの

効用(=情報レン

ト)  2 0.00 0.00 0.00 

      
licd.m(i) 0 –0.20 – – 
 1 –0.70 – – 

誘因両立 制約

のラグランジュ

乗数  2 0.00 – – 

      
pc.m(i) 0 0.00 0.00 – 
 1 0.00 0.00 – 

参加制約のラグ

ランジュ乗数 
 2 –1.00 –1.00 – 

 

 セカンドベスト解をファーストベスト解と比べると、品質 x(i)は、最も効率的なサプライヤー(タイプ 0)

についてのみ一定で、そのほかの非効率なタイプにおいて低下している。これは、メーカーの支払い価

格 w(i)がタイプ 0 についてのみ引き上げられ、他は引き下げられたことと対応する。 

 このように、効率的なサプライヤーが、より非効率的なサプライヤーの振りをすることを阻止するために、

メーカーは彼らの行動を予測して以前よりも気前のよいオファーを出さざるを得なくなる。ただし、このメ

ーカーのオファーは、サプライヤーが自らのタイプを偽っても偽らなくても無差別な水準に決定すること

が最適である(過剰に気前よくなる必要はない)ので、誘因両立制約 licd(i)はタイプ 0, 1 について

のみ有効になる(すなわちそれらのラグランジュ乗数 licd.m(i)はゼロではない)。最も非効率的なサ

プライヤーは(偽るべき、より非効率的なサプライヤーが存在しない以上)、以前と同様に留保効用水準

だけを獲得できるオファーしか得られない。このことは、参加制約のラグランジュ乗数の値がタイプ 2 に

ついてのみ有効(ラグランジュ乗数がゼロではない)であることから確認できる。 

 より効率的なサプライヤー(ここではタイプ 0, 1)が獲得する情報レントは、EQU ブロックの参加制約式

pc(i)の LEVEL と LOWER の差として示される(リスト 2.11 の 137–138 行目)。また、タイプ 0, 1 につ

いては誘因両立制約が等号で成立し、かつ、留保効用 ru を 0 としているので、これらのタイプの効用

はこの情報レントに等しい。(最も非効率なタイプ 2 の効用は、情報レントさえも得られないので留保効

用のまま、すなわち 0 である。)タイプ 1 は、セカンドベストでは価格が低下しているにも関わらず、正の

効用を得ている。これは、セカンドベストではタイプに関する情報の非対称性を利用して、より非効率的

なサプライヤー(タイプ 2)の振りをすることで、ファーストベストのときにはゼロであった情報レントを得るこ

とにより、効用が正に転じたものである。最も効率的なサプライヤーについて効用が増加していることは

言うまでもない。メーカーの効用は、最終的にこうした情報の非対称性に起因する非効率性と、情報レ

ントを通じた収奪を受けることによって、ファーストベストの時よりも低い水準となる。 

 ところで、伊藤(2003)では品質 ix に課されていた単調性の制約(MN)をいったん無視して問題を解き、

その後で解が単調性を満たしていることを確認していた。表 2.7 のセカンドベスト解を見てみると、実際、

効率的なサプライヤーほどより高い品質を提供していることが確認できる。 

 

2.2.5 大域的誘因両立制約を用いたセカンドベスト・モデル (PS3_S_GIC.gms) 

 本項では、前項のモデルの中の、局所的誘因両立制約を大域的誘因両立制約に置き換えたモデル、

すなわち大域的誘因両立制約モデルを解き、両モデルの解の比較を行う。リスト 2.10 の入力ファイル

の理解が十分であれば、新しいモデル(リスト 2.12)の変更点だけを理解しておけばよい。具体的には、

38 行目が(2.4’) に対応したものに書き換えられている。ただし、ここで制約式名 ic の添え字が i と j

の 2 つになっていて、それらはコンマ「,」で区切られていることに注意する。プログラム中の制約式の内

容と(2.5)を比較してわかるように、ic(i,j)の意味は、タイプ i のサプライヤーがタイプ j のサプライ

ヤーの振りをする(ただし、自己のタイプの場合も含む: i=j)ことを阻止するための制約である。 

 

リスト 2.12: 3 タイプのセカンドベスト・モデル–大域的条件の場合(抜粋)  (PS3_S_GIC.gms) 

...  

28 Equation 

29         obj             maker’s utility function 

30         rev(i)         maker's revenue function 

31         pc(i)           participation constraint 

32         ic(i,j)        incentive compatibility constraint; 

33  

34 * Specification of Equations 

35 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

36 rev(i)..b(i) =e= x(i)**(0.5); 

37 pc(i).. w(i)-theta(i)*x(i) =g= ru; 

38 ic(i,j)..w(i)-theta(i)*x(i) =g= w(j)-theta(i)*x(j); 

39  

40 * Setting Lower Bounds on Variables to Avoid Division by Zero 

...  

 

 大域的条件を課した場合でも、SCP が満たされているモデルでは、局所的条件を課したモデルの解

と等しくなる。ただ、制約式が licd(i)から ic(i,j)に変更されているので、そのラグランジュ乗数の

表示が若干変わる(リスト 2.13)。6 つの誘因両立制約のうちで有効なものは、タイプ 0 がタイプ 1 を偽る

もの(145 行目)と、タイプ 1 がタイプ 2 を偽るもの(148 行目)の 2 本である。表現が異なるものの、これら

のラグランジュ乗数の値が表 2.7 に示したものと同じであることがわかる。さらに付言すれば、自分よりも

ひとつだけ効率性の点で劣るタイプを偽る誘因両立制約のみが有効である。これは、そのほかの誘因

両立制約を課さずにこの問題を解いたとしても(=局所的条件 LICD(2.5)のみであったとしても)、同じ品

質や価格といった解を得ることができることを意味する。すなわち、2 つのプログラム(リスト 2.10, 2.11)

が同じ解を導き出すことが、この点でも確認できる。 

リスト 2.13: 3 タイプのセカンドベスト・モデルの解–大域的条件の場合(抜粋) (PS3_S_GIC.lst) 

...  

133 ---- EQU pc  participation constraint 

134  

135      LOWER     LEVEL     UPPER    MARGINAL 

136  

137 0      .        0.522     +INF       .          

138 1      .        0.088     +INF       .          

139 2      .         .         +INF     -1.000       

140  

141 ---- EQU ic  incentive compatibility constraint 

142  

143        LOWER     LEVEL     UPPER    MARGINAL 

144  

145 0.1      .         .        +INF     -0.200       

146 0.2      .        0.346     +INF       .          

147 1.0      .        2.066     +INF       EPS        

148 1.2      .         .        +INF     -0.700       

149 2.0      .        4.478     +INF       .          

150 2.1      .        0.346     +INF       .          

...  

 

2.3 モデルの拡張 

 これまで伊藤のモデルに従ってモデルを構築してきた。とくに、伊藤では解析的に分析をすることを前

提にして、モデルを可能な限り簡単にするようにいくつかの仮定がおかれていた。その第 1 は、

Spence-Mirrlees の単一交差性(SCP)であり、これによって考慮すべき制約式を大幅に少なくしていた。

第 2 は、品質 ix に対して課された単調性制約をいったん無視して問題を解き、事後的に解がこの単調

性を満たしていること確認するという手法を取っていた。より厳密に言うと、単調性が成り立つための条

件を仮定していた(伊藤 1.3.2)。ここでは、これらの条件が満足されない状況を前提として、その際にど

のようにモデルを解くべきかを考える。 

本節では、以上の 2 つの簡単化の仮定に関して、2.3.1 では SCP が成立しないモデルを扱い、

2.3.2 では SCP が成立しているモデルにおける単調性の問題を議論する。加えて、2.3.3 ではタイプの

数が 3 つ以上の場合にどのようにモデルを構築するかについて触れる。 

2.3.1 セカンドベスト・モデル: 大域的誘因両立制約を用いた場合 (PS3_S_SPC.gms) 

 ここでは SCP を満たさない場合を考えてみる。モデルの構造自体は、2.2.5 で提示した大域的誘因

両立モデルと同一である。ただし SCP を満たさない状況を考えるために、サプライヤー iの効用関数 

( )iiii xuwU θ,+= において、関数u を ( ) iiii xxu θθ −=, と特定化してきたものを、 

 ( ) ( )[ ]iiiiii xxu
2

1, θθθθ +−+−=  

に置き換える。(ただし、 10 << ix , 10 << iθ とする。)この関数形を用いることで、 uの ix および iθ

に関する交差偏微分の値 ixu θθ 21−= が、 5.0=iθ より小さければ正、大きければ負と符号が変わる

ので、伊藤の仮定 1.2 は満足されずに SCP は成立しない。ただし、伊藤の仮定 1.3(uがθ の厳密な

減少関数であること)は満足する。15このときのメーカーの効用最大化問題は、 

 ( )[ ]∑ −=
i

iii wxbpUtilmax      (2.1’) 

subject to 

 ( ) 5.0

ii xxb =  i∀       (2.2) 

 ( )[ ] ruxw iiiii ≥+−+− 2
1 θθθ  i∀     (2.3’’) 

 ( )[ ] ( )[ ]jiiijiiiii xwxw
22

11 θθθθθθ +−+−≥+−+−  ji ≠∀   (2.4’) 

となる。 

 プログラム(リスト 2.14, PS3_S_SCP.gms)中では、参加制約(2.3’’)の名前は従来通り pc(i)、大域

的誘因両立制約(2.4’)は ic(i,j)となっている。(プログラム中の sqr(…)は自乗の関数である。)52

行 目 で 、 こ の (2,1’), (2.2), (2.3’’), (2.4’) か ら な る 大 域 的 誘 因 両 立 制 約 モ デ ル の 名 前 を

SB_gic_wo_SCP として定義している。 

 比較のために、本来課すべき大域的条件(2.4’)の代わりに局所的条件(LICD, LICU)、 

                                                      

15 パラメータ に関する順序づけが伊藤 1.3.2 と本稿との間で反対になっているので、満たすべき条件が「増加関

数」ではなく、「減少関数」と反対になっているが、本質的には同じである。 

 ( )[ ] ( )[ ]12

1

2
11 ++ +−+−≥+−+− iiiiiiiiii xwxw θθθθθθ  i∀  

 ( )[ ] ( )[ ]12

1

2
11 −− +−+−≥+−+− iiiiiiiiii xwxw θθθθθθ  i∀  

を課したモデルも構築している。これら 2 つの局所的条件は、licd(i), licu(i)として 43–46 行目

で示されている。53 行目では ic(i,j)の代わりに licd(i), licu(i)を用いた局所的誘因両立制

約モデルの名前を SB_lic_wo_SCP として定義している。 

 

リスト 2.14: SCP を満たさない場合のモデル(抜粋)  (PS3_S_SCP.gms) 

...  

10 * Definition of Parameters 

11 Parameter 

12         theta(i)        efficiency      /0      0.1 

13                                            1       0.4 

14                                            2       0.9/ 

15         p(i)            probability of type 

16                                            /0      0.2 

17                                            1       0.5 

18                                            2       0.3/; 

...  

28 Equation 

...  

32         ic(i,j)         incentive compatibility constraint 

33         licd(i)         incentive compatibility constraint 

34         licu(i)         incentive compatibility constraint; 

35  

36 * Specification of Equations 

37 obj..   Util =e= sum(i, p(i)*(b(i)-w(i))); 

38 rev(i)..b(i) =e= x(i)**(0.5); 

39 pc(i)..  w(i)  -(theta(i)+(1-theta(i)+sqr(theta(i)))*x(i)) 

40              =g= ru; 

41 ic(i,j)..w(i)  -(theta(i)+(1-theta(i)+sqr(theta(i)))*x(i)) 

42     =g=  w(j)  -(theta(i)+(1-theta(i)+sqr(theta(i)))*x(j)); 

43 licd(i)..w(i)  -(theta(i)+(1-theta(i)+sqr(theta(i)))*x(i)) 

44     =g=  w(i+1)-(theta(i)+(1-theta(i)+sqr(theta(i)))*x(i+1)); 

45 licu(i)..w(i)  -(theta(i)+(1-theta(i)+sqr(theta(i)))*x(i)) 

46     =g=  w(i-1)-(theta(i)+(1-theta(i)+sqr(theta(i)))*x(i-1)); 

47  

48 * Setting Lower Bounds on Variables to Avoid Division by Zero 

49 x.lo(i)=0.0001; 

50  

51 * Defining and Solving the Model 

52 Model SB_gic_wo_SCP /obj, rev, pc, ic/; 

53 Model SB_lic_wo_SCP /obj, rev, pc, licd, licu/; 

54  

55 Solve SB_gic_wo_SCP maximizing Util using NLP; 

56 Solve SB_lic_wo_SCP maximizing Util using NLP; 

57 * End of Model 

 

 大域的制約モデルを解いて得られた ix の解は単調性を満たしていない(表 2.8)が、そもそも局所的

条件を用いているわけではないので単調性を満たしている必要はなく、得られた解がそのまま正しい解

である。大域的制約モデルと局所的制約モデルの違いを誘因両立制約の点からみていこう。大域的制

約モデルでは、3 つの場合(タイプ 0 が「タイプ 2」と表明、タイプ 1 が「タイプ 0」と表明、および、タイプ

1 が「タイプ 2」と表明)に、誘因両立制約のラグランジュ乗数がゼロでないので、自らのタイプを偽る誘

因が存在することがわかる。一方、局所的制約モデルでは、この 3 つの場合のうちタイプ 0 が「タイプ

2」と表明する場合を防ぐ制約が欠落している。 (そのほか、タイプ 2 がタイプ 0 を偽ることを防ぐ制約も

欠落しているが、この数値例の結果ではこの制約は有効でないため、この制約式が欠落していることで

解がゆがむことはない。)以上の結果、局所的制約モデルの解は、大域的制約モデルの解から乖離して

誤ったものとなる。局所的制約モデルが誤ったモデルである以上この解自体に意味はないが、必要な

制約式が欠落しているために目的関数の値 Util はより大きくなっている。 

表 2.8: 単一交差性を満たさない場合のモデル間の解の比較 

変数 GAMS プロ

グラム中での

変数名 

サプライヤー

のタイプ 
大域的制約モ

デル 
局 所 的 制 約

モデル 

部品の品質 x(i) 0 0.22 0.30 
  1 0.43 0.51 
  2 0.22 0.16 
     
メーカーの収入 b(i) 0 0.47 0.55 
  1 0.66 0.71 
  2 0.47 0.40 
     

w(i) 0 1.10 1.12 
 1 1.26 1.31 

メーカーの支払

い価格 
  2 1.10 1.04 
     
メーカーの効用 Util  –0.62 –0.61 
     

0 0.80 0.75 
1 0.53 0.52 

サプライヤーの効

用(=情報レント) 

pc.l(i) 
-pc.lo(i) 

2 0.00 0.00 
     

ic.m(i,j) 0 が 1 と表明 0.00 – 
 0 が 2 と表明 –0.40 – 

大域的誘因両立

制約のラグランジ

ュ乗数  1 が 0 と表明 –0.20 – 

  1 が 2 と表明 –0.30 – 

  2 が 0 と表明 0.00 – 

  2 が 1 と表明 0.00 – 
     

licd.m(i) 0 – –0.20 
 1 – –0.70 
 2 – 0.00 
licu.m(i) 0 – 0.00 
 1 – 0.00 

局所的誘因両立

制約のラグランジ

ュ乗数 

 2 – 0.00 
     

pc.m(i) 0 0.00 0.00 
 1 0.00 0.00 

参加制約のラグ

ランジュ乗数 
 2 –1.00 –1.00 

 

2.3.2 単調性が成り立つための十分条件を仮定できないとき (PS3_S_MN.gms) 

 伊藤 1.3 では、SCP を満たすモデルにおいて、品質 ix について単調性が成り立つような十分条件

(伊藤 1.3.2 の仮定 1.4)をおいて問題を簡単化して解くことを考えていた。16しかし、実際の問題を考え

たときにこの十分条件が成り立つとは限らない。17(その一方で、ここで考えていることは単調性が成り立

つための十分条件であるから、この十分条件を満たさない場合であっても単調性が成り立つ場合もあり

得る。)このように単調性が成り立つことを仮定できない場合には、明示的に単調性の条件(MN)(2.6)(あ

るいは、LICU: 11 −− −≥− iiiiii xwxw θθ )を課してモデルを解くことになる。(ここでは、伊藤に倣い、

SCP の成立する場合を考える。) 

 ( )[ ]∑ −=
i

iii wxbpUtilmax      (2.1’) 

subject to 

 ( ) 5.0

ii xxb =  i∀       (2.2) 

 ruxw iii ≥−θ  i∀       (2.3’) 

 11 ++ −≥− iiiiii xwxw θθ  i∀      (2.5) 

 1+≥ ii xx   i∀       (2.6) 

 解析的に解く場合には問題が複雑化するが、数値計算による場合には簡単である。実際、この(2.6)

に相当する制約式 mn(i)を、元のプログラム(PS3_S.gms)に追加して解けばよい(PS3_S_MN.gms): 

 mn(i).. x(i) =g= x(i+1); 

                                                      

16 この仮定は、(伊藤のθ に関する順序づけをそのまま用いて)(i)関数 ( )⋅Φ が準凸であることと、(ii)関数 ( )⋅Φ が

( )θ,x に関して、すべての iについて差分増加であることである。ここで、 

( ) ( ) ( )[ ] ( ) ( )[ ]iiiiiiiiii xuxupFxSx θθθθ ,,1,, 1 −−−=Φ + , 

∑∑ =+=
=−=

i

j j

N

ij ji ppF
01

1 , 

01 =−F  

である。 

17 この条件を満たす可能性は、(事前確率の設定の仕方がまったくのランダムであるとするならば)タイプ数が多くな

るほど急速に小さくなる。この可能性がどの程度のものであるかについては補論で例示する。 

ここで、元のプログラム中のパラメータ設定をそのまま用いた場合には、もともと単調性を満たすことにな

るので、リスト 2.10 に mn(i)を追加して解いた場合にも解は以前とまったく同じで、制約式 mn(i)は

一切有効にはならないことに注意しておく。 

 しかし、この結果は種々のパラメータ設定に依存したものであることに注意する必要がある。たとえば、

事前確率 ip のみを、 

と変えたとき、あるいは、効率性のパラメータ iθ のみを、 

と変えたときには、(LICD に加えて、単調性または LICU の制約を課すことなしには)単調性が成立し

ないので正しい解は得られない。 

 

2.3.3 タイプ数が多いとき (PS10_S.gms) 

 タイプ数が多くなると、追加的仮定無しに解析的に解くことは困難になる一方、数値計算においては、

(実際のパラメータ iθ や事前確率 ip がどのようなものであるべきかという問題を除けば)プログラムの修

正は最小限ですむ。たとえば、10 タイプ・モデル( 9,,...2,1,0=i )の場合には、タイプを表す添え字の

集合を、タイプ数に応じて、 

 Set i type of supplier /0*9/; 

 

と書き換えればよい(リスト 2.15)。ここで「*」(アスタリスク)は連続した数の添え字の入力を簡略化するた

めの記号である。上のようなプログラムの代わりに、 

 Set i type of supplier /0,1,2,3,4,5,6,7,8,9/; 

としても同じである。また、100 タイプ・モデルを解く場合には、最後の「9」の代わりに「99」とすればよい

(ただし、試用版の GAMS を用いた場合には、モデルの大きさの制約のために 100 タイプ・モデルは

解くことができない)。この入力ファイル(PS10_S.gms)の 14–15 行目では、効率性のパラメータ iθ や

事前確率 ip について、タイプ数がいくつになった場合でも数値計算上困難にならない値(一定の率で

逓増する iθ と均一の ip )を自動で計算するようになっているが、これらに代えて個別に値を与えてもよ

い。18 

 

リスト 2.15: 10 タイプ・モデルの入力ファイル(抜粋) (PS10_S.gms) 

...    

7 * Definition of Set 

8 Set     i       type of supplier        /0*9/; 

9 Alias (i,j); 

10 * Definition of Parameters 

11 Parameter 

12         theta(i)        efficiency 

13         p(i)             probability of type; 

14 theta(i)=ord(i)/card(i); 

15 p(i)=1/card(i); 

...    

 

 最後に、これまでに構築してきたモデルと入力ファイルの一覧を表 2.9 に示す。これらのサンプル・プ

ログラムは付録として本稿とは別に提供されている。 

 

表 2.9: 部品調達問題に関するモデル一覧 

入力ファイル名 タイプ

数 
非対称

情報 
誘 因 両

立制約 
単調性

の制約 
伊藤 本稿 備考 

PS2_F_eff.gms 2 なし – – 1.1.2 2.1.2 効率的サプライヤー 
PS2_F_inf.gms 2 なし – – 1.1.2 2.1.3 非効率的サプライヤー 
PS2_F.gms 2 なし – – 1.1.2 2.1.4  

                                                      

18 プログラム中の ord(i)と card(i)は、それぞれ、添え字 i がその集合の中で現れる順番、および、その集合

内の要素の数を表す GAMS の関数である。 

PS2_S.gms 2 あり 局所的 なし 1.1.4 2.1.5  
PS2_F&S.gms 2 なし/ 

あり 
局所的 なし 1.1.2/ 

1.1.4 
2.1.7  

PS3_F.gms 3 なし – – – 2.2.2  
PS3_S.gms 3 あり 局所的 なし 1.3.2 2.2.3  
PS3_S_GIC.gms 3 あり 大域的 – – 2.2.5  
PS3_S_SCP.gms 3 あり 大域的/

局所的 
– – 2.3.1 局所的制約モデルの

解は正しくない 
PS3_S_MN.gms 3 あり 局所的 あり – 2.3.2  
PS10_S.gms 10 あり 局所的 なし – 2.3.3  
PS10_S_MN.gms 10 あり 局所的/

大域的 
あり/ 
なし 

– 補論 単調性制約なしのモ

デルの解は正しくない 

 

3. むすび—結論に代えて 

 本稿では、アドバース・セレクションの基本的なモデルについて、タイプが 2 個の数値モデルと 3 個の

数値モデルを作って解いてきた。仮定された関数形とパラメータに限定されるとはいえ、その下での計

算結果は、伊藤の理論モデルの解析的説明と一致している。すなわち、ファーストベストに比べて、セカ

ンドベストでは最も効率的なタイプのみが品質を維持する。また、彼らは、より非効率的なタイプを偽るこ

とが可能であるために、そこから発生する情報レントを獲得する。一方、最も非効率的なタイプは情報レ

ントを得ることができない。この結果、メーカーの効用は低下する。 

 本稿で、われわれは、モデルを非線形計画問題として定式化して数値計算を行った。こうしたプログラ

ミング・モデル・アプローチは、いくつかの点で有用であることがわかる。第 1 に、セカンドベストの状況

における最も重要な要素は、上で触れたように、情報の非対称性に起因する情報レントである。この情

報レントは、GAMS の出力ファイル中で、制約式に関する計算結果を示す EQU ブロックにある、

LEVEL 値と LOWER 値の差、すなわち、スラックとして陽表的に示される。第 2 に、タイプの数が増加す

ると誘因両立制約が複雑になる。伊藤(2003)では、効用関数に関する SCP の仮定を導入することによ

り、数多い大域的条件を、より少数の局所的条件に置き換えている。コンピュータ・プログラム中では、こ

うした追加的な仮定の有無を簡単にプログラムに反映させることができ、それによって、確かに大域的制

約モデルと局所的制約モデルの間で解が一致すること、あるいは逆に、これらの追加的な仮定が成立

しないときには解が一致しないことを示すことができる。 

 第 3 に、われわれのアプローチでは、解析的方法において用いられる簡単化のための工夫の必要性

GRIPS Policy Research Center Discussion Paper : 10-34



H. Hashimoto, K. Hamada, & N. Hosoe 

Contract Theory: A Programming-Model Approach  Page 47 

は、それほど高くない。すなわち、これらの理論分析でよく用いられるアプローチを離れた場合に元の大

きなモデルをそのまま解いたとしても、追加的費用は気にするほどのものではない。タイプ数をさらに多

く(たとえば 10 個に)したとき、理論的分析は急速に難しくなるが、数値計算上の問題は「追加的に何秒

の時間がかかるか」ということでしかない。第 4 に、数値例による分析に限界があることはいうまでもない

が、しかしながら、補論において行う単調性が満たされる可能性を推量するためのモンテカルロ・シミュ

レーションのような手法は、数値計算モデルにしてはじめて行えることである。 

補論: 単調性が成り立つための十分条件について 

 本文 2.3.2 において単調性が成り立たない場合について考察した。ここでは、単調性に関して以下の

2 点について追加的に検討する。第 1 点は、単調性が成り立つための十分条件を導き出すことである。

第 2 点は、具体的にどの程度の確率で単調性が成り立つのかを示すことである。後者については、2

つの段階で検討する。第 1 は、(A)具体的にどの程度の確率で単調性の十分条件が成り立つかである。

第 2 は、(B)単調性の十分条件の成立・不成立に関わらず、どの程度の確率で解が単調性を満たすか

である。ただし、伊藤 1.3 におけるタイプ iθ の順序づけと、本稿におけるそれが異なっているために、検

討すべき命題そのものについて予備的な考察を A.1 で行う。続いて、A.2 で具体的な確率について検

討する。 

 

A.1 (タイプに関する順序づけを変更した場合の)単調性が成り立つ十分条件 

 第 1 点の単調性が成り立つための十分条件(伊藤 1.3.2 および 1.4)を、タイプ iθ に関する順序づけ

が反対になっている場合について検討する。伊藤における iθ をここでは ( )0<−= −iNi θα 、そのタイプ

の出現確率を ( )iNi pq −= 、品質を ( )ii xy = と表記することにする。 iα は大きいほど効率的という順序

づ け に な る 。 限 界 費 用 は 、 Nθθ << ...0 と す る と 、 iα を 用 い た と き の 表 現 は 、

010 ... θαθαθα −=<<−=<−= − NNiN となる。これによって、伊藤 1.3 で定義されている関数Φ

は、 

 ただし、 { } 11Pr −−
−=

−==≤= ∑ iN

N

iNk

kii FqG αα , 1=NG  

と表現し直すことができ、これが iy と iα に関して差分増加であること(すなわち、 ii αα >+1 ならば、任

意の iy に対して ( ) ( ) 0,, 1 ≥Φ−Φ + iiyiiy yy αα )が単調性を満たす条件であった。(A.1)を iy につい

て微分すると、タイプ iについて、 

 ( ) ( ) ( )ii

i

i
iiiiy

q

G
yby αααα −

−
−+=Φ +1

1
',     (A.2) 

同様に、タイプ 1+i について、 

 ( ) ( ) ( )12

1

1
11

1
', ++

+

+
++ −

−
−+=Φ ii

i

i
iiiiy

q

G
yby αααα    (A.3) 

(A.3)から(A.2)の両辺をそれぞれ差し引くと、 

すべての iについて iα の区間幅が同じ( ( ) ( ) ( ) 01121 >∆=−=−=− ++++ αθθαααα iiiiii )である

状況に限定した場合を考えると、(A.4)の 3 つの括弧内が等しくなるので、伊藤 1.4.2 の単調危険率条

件(monotone hazard rate condition, MHRC)の離散形版に相当する条件、 

が成立すれば、 ( ) ( ) 0,, 1 ≥Φ−Φ + iiyiiy yy αα が言える。 

 ところで、元の iθ の分布関数 iF は、 

 { } ∑
=

=≤=
i

j

jii pF
0

Pr θθ  

 1=NF  

であり、たとえば、 

 0θ については、 00 pF =  

 1θ については、 101 ppF +=  

 … 

であった。一方、 iα の分布関数 iG は、 

 ( )Nθα −=0 については、 100 1 −−=== NN FpqG  

 ( )11 −−= Nθα については、 21101 1 −− −=+=+= NNN FppqqG  

 … 

であるから、 11 −−−= iNi FG となる。したがって、危険率の条件(A.5)は、 

       (A.6) 

と書き換えることができる。これが成立することが単調性が成り立つための十分条件となる。 

 

A.2 単調性が成り立つ確率 

 以上をふまえて、具体的にどの程度の確率で単調性が成り立つのかを検討するために 10 タイプで

iθ の区間幅が一定であるモデル(PS10_S_MN.gms)を作る。第 1 段階の問題は、どの程度の確率で、

単調性が成り立つための十分条件(A.6)が満たされているかを調べるものである。そのため、各タイプ i

の事前確率 ip について(一般には iθ 等の設定と同様に実証的問題であるが)、ここでは区間[0,1]上の

一様分布に従うものとして t=1,000 回のモンテカルロ・シミュレーションを行う。第 2 段階の問題を考え

るため、単調性の制約 mn(i)を課さないモデル(SB_lic)と課したモデル(SB_lic2)の 2 つを作り、そ

れらの解を比較する。(なお、後者は常に正しい解を導く。)両者の解が一致していることは、(単調性の

十分条件の成立・不成立に関わらず)単調性が成立しているこを意味している。したがって、第 2 段階

の問題は、どの程度の確率で 2 つのモデルの解が一致するかということになる。なお、この問題を考察

するモデルの入力ファイル(PS10_S_MN.gms)が込み入っているのは、モンテカルロ・シミュレーション

のために事前確率 p(i)をランダムに決めるためのプログラムが必要であることと、単調性の制約

mn(i)を課したものと課さないものの 2 つを、1 つのモデルで解いているからである。 

第 1 段階の問題(A)まったくランダムに事前確率 ip を設定したときどの程度の確率で MHRC(A.6)

が成立するかについては、この数値例では、1,000 回の試行を行ったときに、MHRC を満足する場合

はただの 1 回(0.1%)である。19ただし、MHRC の成立はあくまでも十分条件であるから、これを満たさ

なくても単調性が成立する、すなわち、上の 2 つのモデルの解が一致するかも知れない。そこで、第 2

段階の問題として、(B)2 つのモデルの解が一致するのはどの程度の頻度であるかをみる。この数値例

では、(B)が発生する場合は 1,000 回の試行で 7 回(0.7%)である。もちろん、この計算は一連の仮定に

基づいているものである。たとえば、タイプ数を減ら(増や)すと、これらの確率が増加(減少)する。同じ設

定で 5 タイプのモデルを考える(プログラム中では、8 行目の「Set  i  type of supplier 

/0*9/」の末尾を「/0*4/」と書き換えればよい)と、(A)が 24%、(B)が 36%程度になる。結論の一般性

については十分に留保する必要があることは承知の上であるが、MHRC(A.6)を仮定して解くことができ

る場合や、単調性を無視して解くことができる場合は、意外と限られたものであると推量される。 

                                                      

19 なお、MHRC(A.6)は iθ と ip の関係だけについての条件であるから、その条件が満足されるかどうかは最大化

問題の内容(たとえば、効用関数の関数型の選択)とは無関係に決まることである。 

Abstract 

This is a study to develop and solve numerical models based on Itoh’s (2003, Ch. 1) “Parts 

Supply Problems” for better understanding the contract theory. In the first part of this 

paper, by following Itoh (2003) we investigate 2- and 3-agent type cases; in the succeeding 

part, by using numerical examples, we examine how likely the simplifying assumptions 

often used in theoretical analysis are to hold. Finally, we demonstrate that we can extend 

these basic models to ones with a much larger number of agent types easily by exploiting 

the merit of our programming–model approach. 

